Lucrarea de laborator nr. 7

I. Scopul lucrării

Rezolvarea ecuațiilor neliniare: Metoda punctului fix, Metoda bisecției, Metoda coardei.

II. Conținutul lucrării

1. Familia de comenzi solve din MAPLE
2. Metoda punctului fix.
3. Metoda bisecției.
4. Metoda coardei.

III. Prezentarea lucrării

III.1. Familia de comenzi **solve** din MAPLE

Spre deosebire de atribuiri, ecuațiile sunt expresii matematice simple care stabilesc relații între anumite variabile şi/sau valori (fără a asocia vreo valoare explicită pentru variabilele conținute). Operatorul folosit este = (reamintim că în cazul atribuirii se utilizează : =). O familie de comenzi care folosesc ecuațiile ca parametrii este familia de comenzi **solve**. Forma generală a comenzii **solve** este:

```
>solve(ecuatie, necunoscuta);
```

sau

```
>solve(ecuatii, neconoscute);
```

MAPLE dispune de comenzi specializate pentru rezolvarea diverselor tipuri de ecuații (rezolvarea ecuațiilor în diverse mulțimi):

fsolve: rezolvă ecuații în virgulă mobilă
isolve: rezolvă ecuații în mulțimea numerelor întregi
msolve: rezolvă ecuații modulo m
rsolve: rezolvă ecuații date prin relații de recurență
dsolve: rezolvă ecuații diferențiale ordinare

linsolve: rezolvă sisteme de ecuații liniare; această comandă aparține pachetului **linalg**, deci înainte de utilizare trebuie încărcat acest pachet sau comanda trebuie apelată sub forma **linalg[linsolve]**. Parametrii acestei comenzi sunt: matricea sistemului A, vectorul termenilor liberi b, şi opțional un al treilea parametru, căruia i se va atribui rangul matricei. Dacă vectorul termenilor liberi se înlocuieşte cu o matrice B, atunci comanda întoarce matricea X care verifică AX = B.

Exemple.

```
> solve(cos(x)+y=9,x);
                    Pi - arccos(y - 9)
>solve({cos(x)+y=9, x+y-Pi/2=9},{x,y});
                    {y = 9, x = 1/2 Pi}
> solve(x^2-2*x+1,x);
                           1, 1
> solve(x^2-2*x+7,x);
```

$$1 + I\sqrt{6}, \quad 1 - I\sqrt{6}$$

```
> dsolve({diff(y(t),t)+t=0, y(1)=2}, y(t));
```

$$y(t) = -\frac{1}{2}t^2 + \frac{5}{2}$$

```
> isolve(3*x+4*y=7);
              {x = 1 - 4 _N1, y = 1 + 3 _N1}
> isolve(3*x+4*y=7,t);
                {x = 1 - 4 t, y = 1 + 3 t}
> msolve({2*x+3*y=1,5*x-2*y=4},5);
                      {y = 3, x = 1}
> rsolve(f(n)=2*f(n-1)+3*f(n-2), f(k));
```

$$\left(\frac{3}{4} f(0) - \frac{1}{4} f(1)\right) (-1)^k - \left(-\frac{1}{4} f(0) - \frac{1}{4} f(1)\right) 3^k$$

```
> with(linalg);
> A:=matrix(3,3,[1,2,3,6-8,7,8,1,0,1]);
```

$$A := \begin{bmatrix} 1 & 2 & 3 \\ -2 & 7 & 8 \\ 1 & 0 & 1 \end{bmatrix}$$

```
> b:=vector(3,[-7,2,3]);
                      b := [-7, 2, 3]

> linsolve(A,b);
```

$$\left[-\frac{34}{3}, -\frac{58}{3}, \frac{43}{3}\right]$$

```
> linsolve(A,b,'r');
```

$$\left[-\frac{34}{3}, -\frac{58}{3}, \frac{43}{3}\right]$$

```
> r;
                              3
> B:=matrix(3,2,[-7,3,2,4,3,5]);
```

$$B := \begin{bmatrix} -7 & 3 \\ 2 & 4 \\ 3 & 5 \end{bmatrix}$$

```
> linsolve(A,B);
```

$$\begin{bmatrix} -\frac{34}{3} & -2 \\ -\frac{58}{3} & -8 \\ \frac{43}{3} & 7 \end{bmatrix}$$

```
> C:=matrix(2,2,[1,2,-5,-10]);
```

$$C := \begin{bmatrix} 1 & 2 \\ -5 & -10 \end{bmatrix}$$

```
> d:=vector(2,[7,-35]);
                    d := [7, -35]

> linsolve(C,d);

              [7 - 2 _t[1], _t[1]]

> b1:=vector(2,[7,-34]);
                    b1 := [7, -34]
> linsolve(C,b1);
>
```

Ultimul sistem (sistemul Cx = b1) este incompatibil. În această situație MAPLE întoarce constanta NULL. Această explică lipsa răspunsului la ultima comandă.

Rezultatele pe care le vom obține, utilizând metodele prezentate în secțiunile următoare vor fi comparate cu rezultatele comenzii **fsolve**. Comanda fsolve admite un set de opțiuni. Specificarea lor nu este obligatorie. Astfel forma generală a comenzii este:

```
> fsolve(ecuatii, necunoscute, optiuni)
```

Opțiunile sunt:

complex: caută rădăcinile ecuației în mulțimea numerelor complexe (în virgulă mobilă).

fulldigits: previne micşorarea numărului de cifre semnificative pe măsură ce se execută calculele; precizia evaluării creşte dar viteză de execuție scade.

maxsols = n: calculează cele mai mici n rădăcini; opțiunea are sens doar pentru ecuațiile polinomiale, pentru că numai pentru acestea se calculează mai multe rădăcini.

interval: caută rădăcinile în intervalul dat; intervalul se specifică sub forma a..b, sau x = a..b, sau {x = a..b, y = c..d, …}; intervalul se consideră închis.

Exemple

```
> fsolve(tan(sin(x))=1,x);
                    2.238253543
> fsolve(x^4-1,x);
                      -1., 1.
> fsolve(x^4-1,x,0..4);

                         1.

> fsolve(x^4-1,x,complex);
               -1., -1. I, 1. I, 1.
> fsolve(x^4-1,x,complex, maxsols=2);
                    -1., -1. I
> Digits:=20;
                   Digits := 20
> fsolve(x^2=2,x=0..infinity);
              1.4142135623730950488
>fsolve(x^2=2,x=0..infinity,fulldigits);
                    1.4142135623730950488
```

III.2. Metoda punctului fix.

Teoremă. Fie E un spaţiu Banach şi S o submulţime închisă a lui E. Fie f : S $\rightarrow$ S o contracţie ($\Leftrightarrow$ există q$\in$ (0,1) astfel încât $\|f(x)-f(y)\| \leq q\|x-y\|$ pentru orice x,y $\in$S). Atunci există şi este unic z punct fix pentru f ($\Leftrightarrow$ f(z)=z).

z este limita unui şir construit după cum urmează:

$x_0 \in$ S dat

$x_n = f(x_{n-1})$, n $\geq$ 1.

Mai mult,

$$\|z-x_n\| \leq \frac{q}{1-q}\|x_n - x_{n-1}\|$$

Algoritm:

Date de intrare:

f (contracţie)

x00 (termenul iniţial al şirului)

eps (eroare ce determină condiţia de oprire: se calculează termenii şirului până la x_N cu proprietatea $\|x_N - x_{N-1}\|$ < eps).

Date de ieşire:

x_N (aproximaţie satisfăcătore (determinată de eps) pentru punctul fix).

x0:=x00;
x1:=f(x0);
cât timp $\|x1-x0\|$ ≥eps **execută**
    x0:=x1;
    x1:=f(x0);

La ieşire x1 este aproximaţie pentru z, punctul fix al lui f.

Faptul că f este contracţie asigură terminarea programului într-un număr finit de paşi. Pentru a evita ciclarea în situaţia în care un utilizator ar încerca folosirea programului pentru o funcţie care nu este contracţie, se poate stabili de la început un număr maxim de paşi ce urmează a fi executaţi. De asemenea se poate afişa la fiecare pas diferenţa dintre termenii consecutivi curenţi. (pentru a observa că şirul nu converge dacă f nu e contracţie). Astfel se poate folosi următoarea variantă a algoritmului:

Date de intrare:

f (contracţie)

x00 (termenul iniţial al şirului)

eps (eroare)

Nmax (număr maxim de termeni ai şirului ce urmează a fi caculaţi)

Condiţia de oprire: se calculează termenii şirului până la x_N cu proprietatea $\|x_N - x_{N-1}\|$ < eps sau N ≥ Nmax).

Date de ieşire:

x_N (dacă f este contracţie, x_N este aproximaţie satisfăcătore - determinată de eps şi Nmax - pentru punctul fix al lui f).

```
x0:=x00;
x1:=f(x0);
n:=1;
cât timp (‖x1 – x0‖ ≥eps) şi (n<Nmax) execută
        x0:=x1;
        x1:=f(x0);
        n:=n+1;
        scrie (x1-x0);
```

Dacă f este contracţie, la ieşirea din ciclu, x1 este aproximaţie pentru z, punctul fix al lui f.

Exemple.

1) Să se rezolve (în mulţimea numerelor reale) ecuaţia următoare aplicând metoda punctului fix

$$x - \cos(x) = 0$$

Notăm g(x) = x – cos(x). Cum cos(x) ∈ [-1, 1], soluţiile ecuaţiei x – cos(x) = 0 se găsesc în intervalul [-1, 1]. Deoarece $g'(x) = 1 + \sin(x) > 0$ pentru x din intervalul [-1, 1], rezultă că g este strict crescătore pe [-1, 1] şi deci ecuaţia

$$x - \cos(x) = 0$$

are cel mult o rădăcină. Cum $g(0)g(\frac{\pi}{2})<0$, ecuaţia x – cos(x) = 0 are exact o rădăcină în intervalul $[0, \frac{\pi}{2}]$. Fie $f: [0, \frac{\pi}{2}] \to [0, \frac{\pi}{2}]$,

$$f(x) = \frac{1}{2}\,(x + \cos(x)).$$

Avem $f'(x) = \frac{1}{2}(1 - \sin(x))$ şi deci

$$\sup_{x\in(0,\pi/2)} |f'(x)| = < 1$$

Putem reprezenta grafic (cu comanda **plot**) funcția x-> 1 - |f '(x) |, şi observa faptul că graficul este deasupra axei Ox. Ca urmare f este contracție pe intervalul $[0, \frac{\pi}{2}]$. Deci soluția ecuației poate fi aflată ca limita şirului $(x_n)_n$, cu

$$x_n = f(x_{n-1}) = \frac{1}{2}\,(x_{n-1} + \cos(x_{n-1})),\ x_0 \in [0, \frac{\pi}{2}].$$

2) Să se rezolve (în mulțimea numerelor reale) ecuația următoare aplicând metoda punctului fix

$$x^3 - x - 1 = 0.$$

Dacă reprezentăm grafic funcția $x \to x^3 - x - 1$, observăm că ecuația $x^3 - x - 1 = 0$ are o singură rădăcină reală, şi că această rădăcină este în intervalul (1,2). Ecuația se poate scrie:

$$x^3 = x+1$$

$$x = \sqrt[3]{x+1}$$

Ca în exemplul anterior se verifică faptul că f, definită prin $f(x) = \sqrt[3]{x+1}$ este contracție pe intervalul (1,2). Deci soluția ecuației poate fi aflată ca limita şirului $(x_n)_n$, cu

$$x_n = f(x_{n-1}) = \sqrt[3]{x+1},\ x_0 \in (1,2).$$

Proceduri MAPLE

```
> pctfix := proc(f, x00, eps)
   local x0, x1;
   x0 := x00;
   x1 := f(x0);
   print(x1, x0, abs(x1 - x0));
   while eps <= abs(x1 - x0) do
      x0 := x1; x1 := f(x0);
      print(x1, x0, abs(x1 - x0))
   od;
   RETURN(x1)
end;

>pctfixN := proc(f, x00, eps, Nmax)
 local x0, x1, n;
 x0 := x00;
 x1 := f(x0);
 n := 1;
 print(x1, x0, abs(x1 - x0));
     while eps<=abs(x1 - x0) and n<Nmax do
```

```
        x0 := x1;
        x1 := f(x0);
        n := n + 1;
        print(x1, x00, abs(x1 - x0))
    od;
    print(`Numar de pasi`, n);
    RETURN(x1)
end;
```

Utilizăm prima procedură pentru rezolvarea ecuației din exemplul 1.

```
>with(plots);
> plot([(x+cos(x))/2,x],x=0..Pi/2);
```




```
> f:=x->(x+cos(x))/2;
```

$$f := x \to \frac{1}{2}x + \frac{1}{2}\cos(x)$$

```
> plot(1-abs(diff(f(x),x)),x=0..Pi/2);
```



```
> pf(f,0.5,0.001);
```

$$0.6887912810,\ 0.0416117825$$
$$0.7304030635,\ 0.0072512438$$
$$0.7376543073,\ 0.0011969453$$
$$0.7388512526,\ 0.0001957026$$
$$0.7390469552$$

```
> fsolve((cos(x)+x)/2=x,x);
```

$$0.7390851332$$

```
> pf(f,Pi/4,10^(-2));
```

$$\frac{\pi}{8}+\frac{\sqrt{2}}{4},\ \frac{\pi}{16}+\frac{\sqrt{2}}{8}-\frac{1}{2}\cos\left(\frac{\pi}{8}+\frac{\sqrt{2}}{4}\right)$$
$$\frac{\pi}{16}+\frac{\sqrt{2}}{8}+\frac{1}{2}\cos\left(\frac{\pi}{8}+\frac{\sqrt{2}}{4}\right)$$

Utilizăm a doua procedură MAPLE pentru funcția

$$x\text{->}\ x^3 + 2x-1$$

Această funcție nu este o contracție. Vom observa că şirul construit nu converge.

```
> g:=(x->x^3+2*x-1);
                    g := x -> x^3  + 2 x - 1
> pctfixN(g,0.1,0.001,5);
                          -.799, .1, .899
             -3.108082399, -.799, 2.309082399
       -37.24078842, -3.108082399, 34.13270602
-51723.84926, -37.24078842, 51686.60847
-.1383797407 10^15, -51723.84926,.1383797406 10^15
                         Numar de pasi, 5
                          .1383797407 10^15
```

Se observă că diferențele dintre termenii consecutivi ai şirului cresc, iar valoarea întoarsă este departe de valoarea reală a rădăcinii ecuației g(x)=x (din cauză că g nu e contracție).

Utilizăm prima procedură pentru rezolvarea ecuației din exemplul 2.

```
>plot(x^3-x-1,x);
```




```
> plot(x^3-x-1,x=-5..5);
```




```
> plot(x^3-x-1,x=-2..2);
```




```
> f:=(x->(x+1)^(1/3));
```

$$f := x \to (x + 1)^{1/3}$$

```
> plot(f(x),x=1..2);
```




```
> plot(1-abs(diff(f(x),x)),x=1..2);
```




```
> pctfix(f,1.1,0.0001);
              1.280579165, 1.1, .180579165
        1.316280308, 1.280579165, .035701143
        1.323113310, 1.316280308, .006833002
        1.324413090, 1.323113310, .001299780
        1.324660046, 1.324413090, .000246956
        1.324706957, 1.324660046, .000046911
                         1.324706957

> fsolve(x^3-x-1,x);
                         1.324717957
```

III.3 Metoda bisecției (metoda înjumătățirii intervalului)

Fie f : [a,b] → R, o funcție continuă cu proprietatea că

$$f(a)f(b) < 0.$$

Atunci există cel puțin o rădăcină $x^* \in (a,b)$ a ecuației f(x)=0. Pentru găsirea rădăcinii se micşorează la fiecare pas intervalul în care funcția îşi schimbă semnul. Metoda bisecției presupune înjumătățirea la fiecare pas a acestui interval. Astfel

- se determină mijlocul $c = \frac{a+b}{2}$ al intervalului (a,b).
- dacă f(c)f(a)<0, atunci se continuă algoritmul cu intervalul [a,c]
- dacă f(c)f(b)<0, atunci se continuă algoritmul cu intervalul [c,b]
- dacă f(c) =0 s-a determinat o rădăcină a ecuației f(x) = 0.

Se construieşte astfel un şir de intervale $(I_n)_n$, $I_n = [a_n, b_n]$, cu lungimea lui I_n egală cu jumătate din lungimea lui I_{n-1}. Fiecare din aceste intervale conține o soluție a ecuației f(x) = 0. Presupunem că se dă o eroare eps>0. Considerăm că c_n mijlocul intervalului I_n este o aproximație satisfăcătoare a soluției ecuației f(x) = 0 dacă lungimea lui I_n este mai mică decât eps.

Algoritm.

Date de intrare:

f continuă

a,b cu f(a)f(b)<0

eps (eroare)

Date de ieşire:

c mijlocul intervalului $I_n = [a_n, b_n]$ cu $| a_n - b_n | < eps$.

cât timp $| a\text{-}b | \geq$ eps **execută**

$c := \frac{a+b}{2}$;

dacă f(c) = 0 **atunci** a := c

b:= c;

altfel **dacă** f(c)f(a)<0 **atunci** b : = c;

altfel a : = c;

Procedură MAPLE

```
bisectie := proc(f, A, B, eps)
 local c, a, b;
 a := A;
 b := B;
 while eps <= abs(b - a) do
   c := (a+b)/2;
   print(c, abs(b - a));
   if f(c) = 0 then a := c; b := c
     else if f(c)*f(a) < 0 then
             b := c
            else a := c
          fi
   fi
 od;
 c := (a+b)/2;
 RETURN(c)
end;
```

Utilizăm această procedură pentru determinarea rădăcinilor reale ecuației:

$$x^8 - 3x + 3 = 0.$$

Reprezentăm grafic funcția

$$x \to x^8 - 3x + 3$$

pentru a localiza rădăcinile.

```
>with(plots)
> plot(x^8-3*x-3,x);
```




```
> plot(x^8-3*x-3,x=-5..5);
```




```
> plot(x^8-3*x-3,x=-2..2);
```



```
> plot(x^8-3*x-3,x=-1.5..1.5);
```



Se observă că ecuaţia are două rădăcini reale. Una în intervalul (-1.5, -1) şi alta în intervalul (1,1.5).

```
>f:=(x-> x^8-3*x-3);
```

$$f := x \to x^8 - 3x - 3$$

```
> bisectie(f,-1.5,-1,10^(-3));
                    -1.250000000, .5
                -1.125000000, .250000000
                -1.062500000, .125000000
                -1.031250000, .062500000
                -1.015625000, .031250000
                -1.007812500, .015625000
                -1.003906250, .007812500
                -1.001953125, .003906250
                -1.000976563, .001953125
                      -1.000488282

> bisectie(f,1,1.5,10^(-3));
                     1.250000000, .5
                 1.375000000, .250000000
                 1.312500000, .125000000
                 1.281250000, .062500000
                 1.265625000, .031250000
                 1.273437500, .015625000
                 1.269531250, .007812500
```

```
1.271484375, .003906250
1.270507813, .001953125
1.270996094
```

III.4. Metoda coardei

Fie $f : [a,b] \to R$, o funcție continuă cu proprietatea că

$$f(a)f(b) < 0.$$

rădăcină a ecuației $f(x)=0$ se caută ca şi în cazul metodei bisecției prin micşorarea la fiecare pas a intervalului în care funcția îşi schimbă semnul. Metoda coardei presupune:

- se determină punctul c în care coarda AB intersectează axa Ox, unde $A(a,f(a))$ şi $B(b,f(b))$.
- dacă $f(c)f(a)<0$, atunci se continuă algoritmul cu intervalul $[a,c]$
- dacă $f(c)f(b)<0$, atunci se continuă algoritmul cu intervalul $[c,b]$
- dacă $f(c) =0$ s-a determinat o rădăcină a ecuației $f(x) = 0$.

Se construieşte astfel un şir de intervale $(I_n)_n$, $I_n = [a_n, b_n]$, cu lungimea lui I_n strict mai mică decât lungimea lui I_{n-1}. Fiecare din aceste intervale conține o soluție a ecuației $f(x) = 0$. Presupunem că se dă o eroare eps>0. Considerăm că c_n
din $I_n = [a_n, b_n]$ este o aproximație satisfăcătoare a soluției ecuației $f(x) = 0$ dacă $| c_n - c_{n-1} | <$ eps.



Să determinăm coordonatele punctului C de intersecție a dreptei AB cu axa Ox, unde $A(a,f(a))$ şi $B(b,f(b))$.

Ox: $y = 0$

AB: $\frac{x-a}{b-a} = \frac{y-f(a)}{f(b)-f(a)}$

Dacă C(c,0) =AB $\cap$ Ox, atunci

$$c = a - \frac{b-a}{f(b)-f(a)} f(a)$$

Algoritmul se bazează pe următoarea teoremă:

Teoremă. Fie f : [a, b] $\to$ **R** o funcție de două ori derivabilă cu proprietățile: $f'(x) \neq 0$, $f''(x) \neq 0$, oricare ar fi $x \in [a, b]$ şi $f(a)f(b) < 0$. Atunci unica soluție a ecuației f(x) = 0 poate fi obținută ca limită a şirului strict monoton din [a, b] definit prin:

$x_0 = a$, $x_n = x_{n-1} - \frac{f(x_{n-1})}{f(x_{n-1})-f(b)}(x_{n-1}-b)$, dacă $f(a)f''(a)<0$

şi

$x_0 = ab$, $x_n = x_{n-1} - \frac{f(x_{n-1})}{f(x_{n-1})-f(a)}(x_{n-1}-a)$, dacă $f(b)f''(b)<0$

Dacă $m_1 > 0$, $M_1 > 0$ şi $m_1 \leq |f'(x)| \leq M_1$, atunci unica soluției, x*, a ecuației satisface inegalitățile:

$$|x^*-x_n| \leq \frac{|f(x_n)|}{m_1}$$

$$|x^*-x_n| \leq \frac{M_1 - m_1}{m_1}|x_n - x_{n-1}|$$

Algoritm.

Date de intrare:

f continuă

a,b cu $f(a)f(b)<0$

eps (eroare)

Date de ieşire:

$$c = a_n - \frac{b_n - a_n}{f(b_n)-f(a_n)} f(a_n)$$

unde $I_n = [a_n, b_n]$ are proprietatea $|c_n-c_{n-1}|<$eps.

v: = a ; c: = b;

cât timp | v-c | ≥ eps **execută**
v: = c;
c: = a - $\frac{b-a}{f(b)-f(a)}$ f(a);
dacă f(c) = 0 **atunci stop**
altfel **dacă** f(c)f(a)<0 **atunci** b : = c;
altfel a : = c;

Procedură MAPLE

```
> coarda := proc(f, A, B, eps)
    local c, v, a, b;
    a := A;
    b := B;
    c := a;
    v := b;
    while eps <= abs(v - c) do
     v := c;
     c := a - (b - a)*f(a)/(f(b) - f(a));
     print(c, abs(v - c));
     if f(c) = 0 then RETURN(c) fi;
     if f(c)*f(a) < 0 then
        b := c else a := c fi
 od;
 c := a - (b - a)*f(a)/(f(b) - f(a));
 RETURN(c)
end;
```

Aplicăm această procedură pentru determinarea rădăcinilor reale ale ecuației:

$$x^8 - 3x + 3 = 0.$$

În secțiunea precedentă rădăcinile au fost localizate în intervalele (+1.5,-1) şi (1,1.5).

```
> coarda(f,1,1.5,10^(-3));
```

```
1.108089850, .108089850
```

```
                 1.179673278, .071583428
                 1.222375759, .042702481
                 1.245952470, .023576711
                 1.258349697, .012397227
                 1.264690610, .006340913
                 1.267886414, .003195804
                 1.269484923, .001598509
                 1.270281421, .000796498
                        1.270677536
> coarda(f,-1.5,-1,10^(-3));

                -.9808641053, .5191358947
                -.9046185425, .0762455628
                -.8861367350, .0184818075
                -.8815645598, .0045721752
                -.8804316102, .0011329496
                -.8801508169, .0002807933
                       -.8800812217

> fsolve(f(x),x);
                -.8800582880, 1.271068437
```

Probleme propuse

Să se determine rădăcinile ecuațiilor

1) $e^{-x} - x = 0$
2) $x^2 - 4 \sin(x) = 0$